Supporting Information

A delayed decoupling methyl-TROSY pulse scheme for atomic resolution studies of folded proteins and RNA in condensates


























/* hsqc_800_lek_delaydecW_cp

  Used to record 13C 1H HSQC, with options for 15N, 13C and 2H decoupling during t1

  Written by LEK and RM on Nov 26

  Modified by LEK on Feb 12, 2015 to correct GRAMP/GRAD issue

  Modified by LEK on July 19, 2016 to include coherence transfer gradients
   that are useful when using ddx4 samples where only a small fraction of
   material is 13C labeled

  Modified by LEK on April 26 to ensure that gradients do not interfer with 15N decoupling

  Delayed decouping HSQC !! (not hmqc) with coherence transfer gradients - USED for CAPRIN1 samples
     if no CAPRIN1 then dont use coherence transfer; written by LEK on Dec 4, 2023
*/

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

;Define phases
#define zero ph=0.0
#define one ph=90.0
#define two ph=180.0
#define three ph=270.0
 
;Define Pulses
define pulse pwh
       "pwh=p1"             ; 1H hard pulse at power pl1
define pulse pwc
       "pwc=p2"             ; 13C hard pulse at power pl2
define pulse pwc_ad         ; adiabatic pulse for inept transfers
       "pwc_ad = p23"
define pulse pwc_reb        ; reburp pulse for gradients during t1
       "pwc_reb = p24"
define pulse pwd
       "pwd=p4"             ; 2H pulse at power pl4

;Define delays

"in0=inf1/2"
"d11=30m"

define delay taua
       "taua=d5"            ; < 1/(4JCH) 1.8 ms 

define delay tau1

define delay hscuba
  "hscuba=30m"

;define flags
;f1180    ; set zgoptns -Df1180

#ifdef f1180
  "d0=(in0)/2"
#else
  "d0=(0.2u)/2"
#endif

#ifdef OneDarray
#else
define loopcounter ni
  "ni=td1/2"
#endif

;Ndec    ; set zgoptns -DNdec
;COdec   ; set zgoptns -DCOdec
;Ddec    ; set zgoptns -DDdec

/* Assign cnsts to check validity of parameter changes */
#ifdef fsat
   "cnst10 = plw10" ; tsatpwr - set to max of 0.00005W
#endif
   "cnst21=plw21" ; dpwr pl21 - set max at 4.5W
#ifdef COdec
   "cnst22=plw22" ; dpwrco pl22/sw22 - set max at 8.0W
#endif

"cnst23=spw23" ; adiabatic 13C pulse
"cnst24=spw24" ; reburp 13C pulse

#ifdef Ndec
   "cnst31=plw31" ; dpwr2 pl31 - set max at 5.8W
#endif

#ifdef Ddec
   "cnst4=plw4"   ; dpwr3 pl4 - set max at 10.5W
   "cnst41=plw41" ; dpwr3D pl41 - set max at 1.5W
#endif

; "acqt0=0"  ; select 'DIGIMOD = baseopt' to execute

"d20 = 1s/(cnst25*2.0) - 1.0m"  ; cnst25 is J_methyl

1 ze

; check validity of parameters

#ifdef fsat
  if "cnst10 > 0.001"
  {
   2u
   print "error: presat power pl10 is too large"
   goto HaltAcqu
  }
#endif

  if " cnst21 > 6.0"
  {
   2u
   print "error: dpwr pl21 too large"
   goto HaltAcqu
  }

#ifdef COdec
  if " cnst22 > 1.0"
  {
   2u
   print "error: dpwrco pl22 too large"
   goto HaltAcqu
  }
#endif

  if " cnst23 > 71.0"
  {
   2u
   print "error: power for adiabatic 13C 180 is too large"
   goto HaltAcqu
  }

if "pwc_reb > 30u" {
  if " cnst24 > 80.0"
  {
   2u
   print "error: power for reburp 13C 180 is too large"
   goto HaltAcqu
  }
}

#ifdef Ndec
  if " cnst31 > 4.0"
  {
   2u
   print "error: dpwr2 pl31 too large"
   goto HaltAcqu
  }
#endif

  if "pwc > 20u"
  {
   2u
   print "error: pwc too large < 20 us"
   goto HaltAcqu
  }

  if "pwc_reb > 2m"
  {
   2u
   print "error: pwc_reb too large < 2m"
   goto HaltAcqu
  }

  if "pwc_ad > 500u"
  {
   2u
   print "error: pwc_ad too large < 500 us"
   goto HaltAcqu
  }

#ifdef Ddec
  if "cnst4 > 15.5"
  {
   2u
   print "error: dpwr3 pl4 too large"
   goto HaltAcqu
  }

  if "cnst41 > 4.0"
  {
   2u
   print "error: dpwr3D pl41 too large"
   goto HaltAcqu
  }

;   d11 LOCKDEC_ON ; Not required for AvanceIII-HD 
  50u LOCKH_ON
  d11 H2_PULSE
  2u pl41:f4      ; set power pl4 for 2H flipback pulses
#endif

2 d11 do:f2 

#ifdef Ddec    /* D decoupling */
d11 H2_LOCK    ; put lock channel in lock mode
6m LOCKH_OFF   ; turn off lock hold

#ifdef fsat                           /* zgoptn -Dfsat */
  4u pl10:f1                          ; power(tsatpwr) for presaturation
  d1 cw:f1 zero                       ; Hcw(d1)x
  4u do:f1                            ; cw off
  2u pl1:f1                           ; power(tpwr)

#ifdef fscuba                   /* Scuba pulse sequence */
   hscuba                          ; delay(hscuba)
    (pwh zero):f1                  ; H 90x180y90x
    (pwh*2 one):f1
    (pwh zero):f1
   hscuba                          ; delay(hscuba)
#endif                           /* end fscuba */

#else                             /* if fsat is no */
 2u pl1:f1                        ; power(tpwr)
 d1                               ; delay(d1)
#endif                            /* end if fsat */

50u  LOCKH_ON
15u H2_PULSE

#else    /* no D decoupling */

#ifdef fsat                           /* zgoptn -Dfsat */
  4u pl10:f1                          ; power(tsatpwr) for presaturation
  d1 cw:f1 zero                       ; Hcw(d1)x
  4u do:f1                            ; cw off
  2u pl1:f1                           ; power(tpwr)

#ifdef fscuba                   /* Scuba pulse sequence */
   hscuba                          ; delay(hscuba)
    (pwh zero):f1                  ; H 90x180y90x
    (pwh*2 one):f1
    (pwh zero):f1
   hscuba                          ; delay(hscuba)
#endif                           /* end fscuba */

#else                             /* if fsat is no */
 2u pl1:f1                        ; power(tpwr)
 d1                               ; delay(d1)
#endif                            /* end if fsat */
#endif                            /* end of D_decoupling */

  2u pl1:f1
  2u pl2:f2
  2u pl31:f3

#ifdef Ddec
 20u UNBLKGRAMP          ; dly 20u, unblank gradients
#else
 20u UNBLKGRAD           ; dly 20u, unblank gradients and lock on hold
#endif

  (pwc zero):f2   ; C90x - To destroy 13C Boltzman magnetization

  2u 
  (p50:gp0)       ; gradient 0
  d16

; This is the real start

"tau1 = d0"

if "tau1 < 0.2u" {
   "tau1 = 0.2u"
}

  (pwh zero):f1                             ; H90x

  2u
  p51:gp1                                  ; gradient 1
  d16

 "DELTA = taua - 2.0u - p51 - d16 - pwc_ad*0.5"
  DELTA                                    ; delay 1/4JCH

  (center(pwh*2 ph26):f1 (pwc_ad:sp23 ph26):f2)  ; H180x,C180x

 "DELTA = taua - 2.0u - p51 - d16 - pwc_ad*0.5"
  DELTA                                    ; delay 1/4JCH

  2u
  p51:gp1                                  ; gradient 1
  d16

  (pwh one):f1                             ; H90y

  2u
  p52:gp2                                  ; gradient 2
  d16                   

/*  If zgoptn -DDdec turn on 2H dec */

# ifdef Ddec
  2u pl4:f4             ; dly 2u,  set pwr pl4 dpwr3
  (pwd one):f4          ; 2H 90(y)
  2u pl41:f4            ; dly 2u,  set pwr pl41 dpwr3D
  (2u cpd4 zero):f4     ; Turn on 2H decoupling - cpd4, phase x
# endif

/*  If zgoptn -DNdec turn on 15N dec */
# ifdef Ndec
  2u pl31:f3   ; set power pl31 for 15N decoupling
  (2u cpds3 ph26):f3  ; Turn on 15N decoupling, cpdprg3 - waltz16
# endif

  2u pl2:f2
  (pwc ph3):f2                              ; C90ph3

/* If zgoptn -DCOdec turn on off-resonance CO dec */

# ifdef COdec
  2u pl22:f2
  (2u cpds8 ph26):f2  ; Turn on CO dec, cpdprg8, sync mode 
# else
  2u
  2u
#endif

"DELTA = 10u - 4u"
DELTA

  2u
  p55:gp5*-1.0                       ; coherence transfer gradient 5
  d16                   

  (pwc_reb:sp24 ph26):f2  ; reburp C180x

  2u
  p55:gp5                            ; coherence transfer gradient 5
 "DELTA = d16 - pwh - 1u - pwh*2 - 1u - pwh"
  DELTA

  tau1                               ; t1/2

  (pwh ph26):f1                  ; H90x
  1u
  (pwh*2 ph27):f1                ; H180y
  1u
  (pwh ph26):f1                  ; H90x

   tau1

# ifdef COdec
  2u do:f2   ; Turn off CO decoupling 
  2u pl2:f2  ; set 13C high power
#else
  2u
  2u
# endif

# ifdef Ndec  
  2u
  2u do:f3    ; Turn off 15N decoupling on channel 3
#else
  2u
  2u
# endif

  2u pl2:f2
  (pwc ph5):f2                 ; C90ph5 

# ifdef Ddec
  2u do:f4              ; Turn off 2H decoupling on channel 4
  2u pl4:f4             ; dly 2u,  set pwr pl4 dpwr3
  (pwd three):f4        ; 2H 90(-y)
# endif

  4u
  p53:gp3                       ; gradient 3
  d16 

 (pwh zero):f1                 ; H90x

 "DELTA = 0.5m - pwh*2.0/PI + de" 
  DELTA                                    ; delay 1/4JCH

  (center(pwh*2 ph26):f1 (pwc*2.0 ph26):f2)  ; H180x,C180x

 "DELTA = 0.5m - 2u - p56 - d16 - 4u - 2u"
  DELTA                                    ; delay 1/4JCH

  2u 
  p56:gp6*EA                               ; gradient 6 CHT
  d16 

#ifdef Ddec
  4u BLKGRAMP
#else
  4u BLKGRAD
#endif

  2u pl21:f2               ; lower power for 13C decoupling

  go=2 ph31 cpds2:f2        ; acquire fid with syncronous 13C decoupling during aq
  d11 do:f2 mc #0 to 2
  F1EA(calgrad(EA),calph(ph3,+180) & calph(ph31, +180) & caldel(d0, +in0))

#ifdef Ddec
d11 H2_LOCK
d11 LOCKH_OFF
; d11 LOCKDEC_OFF ; use statement for earlier hardware
#endif

HaltAcqu, 1m
exit

ph3=0 2 
ph5=0 0 2 2
ph31=0 2 2 0
ph26=0
ph27=1
ph28=2
ph29=3

;d1 : repetition delay
;d5 : taua ~1/4JCH
;d11 : delay for disk i/o, 30ms
;d16 : gradient recovery delay, 200us
;pl1 : tpwr - power level for pwh 
;pl2 : dhpwr - power level for hard 13C pulse pwc 
;pl21 : dpwr - power level for  13C decoupling cpd2
;pl22 : dpwrcodec - power level for cos modulated seduce
;sp22 : cos modulated seduce decoupling pattern 
;sp23 : power level for pwc_ad
;p23  : pwc_ad - adiabatic pulses for INEPT transfer
;p24  : pwc_reb - reburp pulses during t1
;pl31 : dpwr2 - power level for 15N cpd3
;pl4 : dpwr3 - power level for 2H flipback pulses
;pl41 : dpwr3D - power level for 2H cpd4
;p1 : pwh
;p2 : pwc
;p22 : pwco90 (seduce1 dec) pattern length us @ pl22 for CO decoupling
;p31 : pwn at dpwr2 for 15N decoupling during 2*TC(~29ms)
;p4 : pwd at dpwr3 for 2H flipback pulses
;p41 : pwddec at dpwr3D for 2H decoupling
;ni : td1/2 number of complex points in t1
;cpd2 : 13C decoupling according to program defined by cpdprg2
;cpd3 : 15N decoupling according to program defined by cpdprg3
;cpd4 : 2H decoupling according to program defined by cpdprg4
;cpd8 : 13CO decoupling according to program defined by cpdprg8
;pcpd2:  13C 90 degree pulse at pl21 for cpd2
;pcpd3:  15N 90 degree pulse at pl31 for cpd3
;pcpd4:  2H 90 degree pulse at pl41 for cpd4
;pcpd8:  seduce1 decoupling pattern length for cpd8
;spnam22 : file name for CO decoupling
;spnam23 : file name for chirp pulse during inepts
;spnam24 : file name for reburp pulse during t1
;cnst25 : actual JCH value 125Hz for ILV 143 Hz for CH3-O-sugar
;zgoptns : Df1180,DNdec,DCOdec,DDdec,Dfsat,DOneDarray,Dfscuba
